Rules for integrands of the form $\left(a x^q + b x^n + c x^{2 n-q}\right)^p$

1: $\int (a x^n + b x^n + c x^n)^p \, dx$

- **Rule:**

$$\int (a x^n + b x^n + c x^n)^p \, dx \longrightarrow \int ((a + b + c) x^n)^p \, dx$$

- **Program code:**

```
Int[(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  Int[((a+b+c)*x^n)^p,x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n,q] && EqQ[r,n]
```

2: $\int \left(a x^q + b x^n + c x^{2 n-q}\right)^p \, dx$ when $q < n \land p \in \mathbb{Z}$

- **Rule: If** $q < n \land p \in \mathbb{Z}$, **then**

$$\int \left(a x^q + b x^n + c x^{2 n-q}\right)^p \, dx \longrightarrow \int x^{p q} \left(a + b x^{n-q} + c x^{2 (n-q)}\right)^p \, dx$$

- **Program code:**

```
Int[(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  Int[x^(p*q)*(a+b*x^(n-q)+c*x^(2*(n-q)))^p,x] /;
FreeQ[{a,b,c,n,q},x] && EqQ[r,2*n-q] && PosQ[n-q] && IntegerQ[p]
```

3: $\int\sqrt{a\,x^q+b\,x^n+c\,x^{2\,n-q}}\,dx$ when $q<n$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x\frac{\sqrt{a\,x^q+b\,x^n+c\,x^{2\,n-q}}}{x^{q/2}\sqrt{a+b\,x^{n-q}+c\,x^{2\,(n-q)}}}==0$
- **Rule: If** $q<n$**, then**

$$\int\sqrt{a\,x^q+b\,x^n+c\,x^{2\,n-q}}\,dx\;\longrightarrow\;\frac{\sqrt{a\,x^q+b\,x^n+c\,x^{2\,n-q}}}{x^{q/2}\sqrt{a+b\,x^{n-q}+c\,x^{2\,(n-q)}}}\int x^{q/2}\sqrt{a+b\,x^{n-q}+c\,x^{2\,(n-q)}}\,dx$$

- **Program code:**

```
Int[Sqrt[a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.],x_Symbol] :=
  Sqrt[a*x^q+b*x^n+c*x^(2*n-q)]/(x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))])*
    Int[x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))],x] /;
FreeQ[{a,b,c,n,q},x] && EqQ[r,2*n-q] && PosQ[n-q]
```

4. $\int\frac{1}{\sqrt{a\,x^q+b\,x^n+c\,x^{2\,n-q}}}\,dx$ when $2<n\,\wedge\,b^2-4\,a\,c\neq0$

1: $\int\frac{1}{\sqrt{a\,x^2+b\,x^n+c\,x^{2\,n-2}}}\,dx$ when $2<n\,\wedge\,b^2-4\,a\,c\neq0$

- **Derivation: Integration by substitution**
- **Basis:** $\frac{1}{\sqrt{a\,x^2+b\,x^n+c\,x^{2\,n-2}}}==-\frac{2}{n-2}\,\text{Subst}\left[\frac{1}{4\,a-x^2},\,x,\,\frac{x\,(2\,a+b\,x^{n-2})}{\sqrt{a\,x^2+b\,x^n+c\,x^{2\,n-2}}}\right]\partial_x\frac{x\,(2\,a+b\,x^{n-2})}{\sqrt{a\,x^2+b\,x^n+c\,x^{2\,n-2}}}$
- **Rule: If** $2<n\,\wedge\,b^2-4\,a\,c\neq0$**, then**

$$\int\frac{1}{\sqrt{a\,x^2+b\,x^n+c\,x^{2\,n-2}}}\,dx\;\longrightarrow\;-\frac{2}{n-2}\,\text{Subst}\left[\int\frac{1}{4\,a-x^2}\,dx,\,x,\,\frac{x\left(2\,a+b\,x^{n-2}\right)}{\sqrt{a\,x^2+b\,x^n+c\,x^{2\,n-2}}}\right]$$

Program code:

```
Int[1/Sqrt[a_.*x_^2+b_.*x_^n_.+c_.*x_^r_.],x_Symbol] :=
  -2/(n-2)*Subst[Int[1/(4*a-x^2),x],x,x*(2*a+b*x^(n-2))/Sqrt[a*x^2+b*x^n+c*x^r]] /;
FreeQ[{a,b,c,n,r},x] && EqQ[r,2*n-2] && PosQ[n-2] && NeQ[b^2-4*a*c,0]
```

2: $\int \frac{1}{\sqrt{a x^q + b x^n + c x^{2 n-q}}} \, dx$ when $q < n$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{x^{q/2} \sqrt{a+b x^{n-q}+c x^{2 (n-q)}}}{\sqrt{a x^q+b x^n+c x^{2 n-q}}} == 0$

Rule: If $q < n$, then

$$\int \frac{1}{\sqrt{a x^q + b x^n + c x^{2 n-q}}} \, dx \to \frac{x^{q/2} \sqrt{a + b x^{n-q} + c x^{2 (n-q)}}}{\sqrt{a x^q + b x^n + c x^{2 n-q}}} \int \frac{1}{x^{q/2} \sqrt{a + b x^{n-q} + c x^{2 (n-q)}}} \, dx$$

Program code:

```
Int[1/Sqrt[a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.],x_Symbol] :=
  x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))]/Sqrt[a*x^q+b*x^n+c*x^(2*n-q)]*
    Int[1/(x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))]),x] /;
FreeQ[{a,b,c,n,q},x] && EqQ[r,2*n-q] && PosQ[n-q]
```

5: $\int \left(a x^q + b x^n + c x^{2 n-q}\right)^p \, dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land p > 0 \land p (2 n - q) + 1 \neq 0$

Derivation: Generalized trinomial recurrence 1b with $m = 0$, $A = 1$ and $B = 0$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land p > 0 \land p (2 n - q) + 1 \neq 0$, then

$$\int \left(a x^q + b x^n + c x^{2 n-q}\right)^p dx \to \frac{x \left(a x^q + b x^n + c x^{2 n-q}\right)^p}{p (2 n - q) + 1} + \frac{(n - q) p}{p (2 n - q) + 1} \int x^q \left(2 a + b x^{n-q}\right) \left(a x^q + b x^n + c x^{2 n-q}\right)^{p-1} dx$$

Program code:

```
Int[(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x*(a*x^q+b*x^n+c*x^(2*n-q))^p/(p*(2*n-q)+1) +
  (n-q)*p/(p*(2*n-q)+1)*
    Int[x^q*(2*a+b*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p-1),x] /;
FreeQ[{a,b,c,n,q},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && NeQ[p*(2*n-q)+1,0]
```

6: $\int \left(a x^q + b x^n + c x^{2 n-q}\right)^p \, dx$ when $q < n \land p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land p < -1$

Derivation: Generalized trinomial recurrence 2b with $m = 0$, $A = 1$ and $B = 0$

Rule: If $q < n \land p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land p < -1$, then

$$\int \left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p\,dx \longrightarrow$$

$$-\frac{x^{-q+1}\left(b^2 - 2\,a\,c + b\,c\,x^{n-q}\right)\left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^{p+1}}{a\,(n-q)\,(p+1)\left(b^2 - 4\,a\,c\right)} + \frac{1}{a\,(n-q)\,(p+1)\left(b^2 - 4\,a\,c\right)}\cdot$$

$$\int x^{-q}\left((p\,q+1)\left(b^2 - 2\,a\,c\right) + (n-q)\,(p+1)\left(b^2 - 4\,a\,c\right) + b\,c\,(p\,q + (n-q)\,(2\,p+3) + 1)\,x^{n-q}\right)\left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^{p+1}\,dx$$

Program code:

```
Int[(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  -x^(-q+1)*(b^2-2*a*c+b*c*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(a*(n-q)*(p+1)*(b^2-4*a*c)) +
  1/(a*(n-q)*(p+1)*(b^2-4*a*c))*
    Int[x^(-q)*((p*q+1)*(b^2-2*a*c)+(n-q)*(p+1)*(b^2-4*a*c)+b*c*(p*q+(n-q)*(2*p+3)+1)*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1),x] /
FreeQ[{a,b,c,n,q},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && LtQ[p,-1]
```

7: $\int \left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p\,dx$ when $q < n \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\left(a\,x^q+b\,x^n+c\,x^{2\,n-q}\right)^p}{x^{p\,q}\left(a+b\,x^{n-q}+c\,x^{2\,(n-q)}\right)^p} == 0$

Rule: If $q < n \land p \notin \mathbb{Z}$**, then**

$$\int \left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p\,dx \longrightarrow \frac{\left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p}{x^{p\,q}\left(a + b\,x^{n-q} + c\,x^{2\,(n-q)}\right)^p}\int x^{p\,q}\left(a + b\,x^{n-q} + c\,x^{2\,(n-q)}\right)^p\,dx$$

Program code:

```
Int[(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  (a*x^q+b*x^n+c*x^(2*n-q))^p/(x^(p*q)*(a+b*x^(n-q)+c*x^(2*(n-q)))^p)*
    Int[x^(p*q)*(a+b*x^(n-q)+c*x^(2*(n-q)))^p,x] /;
FreeQ[{a,b,c,n,p,q},x] && EqQ[r,2*n-q] && PosQ[n-q] && Not[IntegerQ[p]]
```

X: $\int \left(a\,x^q+b\,x^n+c\,x^{2\,n-q}\right)^p\,dx$

- **Rule:**

$$\int \left(a\,x^q+b\,x^n+c\,x^{2\,n-q}\right)^p\,dx \;\longrightarrow\; \int \left(a\,x^q+b\,x^n+c\,x^{2\,n-q}\right)^p\,dx$$

- **Program code:**

```
Int[(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  Unintegrable[(a*x^q+b*x^n+c*x^(2*n-q))^p,x] /;
FreeQ[{a,b,c,n,p,q},x] && EqQ[r,2*n-q]
```

S: $\int \left(a\,u^q+b\,u^n+c\,u^{2\,n-q}\right)^p\,dx$ when $u == d+e\,x$

- **Derivation: Integration by substitution**

- **Rule: If $u == d+e\,x$, then**

$$\int \left(a\,u^q+b\,u^n+c\,u^{2\,n-q}\right)^p\,dx \;\longrightarrow\; \frac{1}{e}\,\text{Subst}\left[\int \left(a\,x^q+b\,x^n+c\,x^{2\,n-q}\right)^p\,dx,\,x,\,u\right]$$

- **Program code:**

```
Int[(a_.*u_^q_.+b_.*u_^n_.+c_.*u_^r_.)^p_,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a*x^q+b*x^n+c*x^(2*n-q))^p,x],x,u] /;
FreeQ[{a,b,c,n,p,q},x] && EqQ[r,2*n-q] && LinearQ[u,x] && NeQ[u,x]
```